

Report On
Floating Pt Square-Root Implementation
submitted to
Dr. Gaurav Trivedi
Ankita Tiwari
Meenali Janveja
by

Abhishek Verma (214102401)
Chaitanya Tejaswi (214102408)
Pawan Kumar (214102412)

April 26, 2022



Department of Electronics and Electrical Engineering
Indian Institute of Technology, Guwahati
Guwahati-781039, India

Aim

Perform post-synthesis simulations & functionality testing for Floating Pt Square-Root Implementation.

Theory

Introduction


With the fast development of sub-micron technology, the density of silicon grows rapidly and the cost of hardware decreases, more and more functionalities are transferred into hardware to realize. In order to meet the ever increasing demand in high performance applications including scientific computations, digital signal processing, computer graphics, multimedia, etc. the performance of square root computation is becoming more and more important. In this design we have used Newton Raphson method to implement square root of a number. The iteration of NR method results in a doubled accuracy which leads to faster execution time. Several methods are commonly employed for the computation of floating-point square root. For example:


- Digital recurrence, it's simple, easy to implement, but the latency is long, especially for large operand sizes.
- Functional iteration, such as NR and Goldschmidt iteration algorithms. It requires more hardware cost, and the result is not fully accurate, but it is fast, scalable, easy to pipeline, and has high precision.
- Very high radix arithmetic, it is fast, but very complicated to implement.
- Table look up, it is simple, fast, with big hardware cost and bad extensibility. 5) Variable latency, it is fast with more hardware cost.

IEEE754 Representation of Floating Pt Numbers

There are several ways to represent floating point number but IEEE 754 is the most efficient in most cases. IEEE 754 has 3 basic components:

- The Sign of Mantissa: This is as simple as the name. 0 represents a positive number while 1 represents a negative number.
- The Biased exponent: The exponent field needs to represent both positive and negative exponents. A bias is added to the actual exponent in order to get the stored exponent.
- The Normalised Mantissa: The mantissa is part of a number in scientific notation or a floating-point number, consisting of its significant digits. Here we have only 2 digits, i.e. 0 and 1. So a normalised mantissa is one with only one 1 to the left of the decimal.





Newton/Raphson Approximation

$$x_{i+1} = x_i - \frac{f(x_i)}{f'(x_i)}$$

This is the NR formula, where x_i is the value at the i th iteration, $f(x_i)$ is the function value at x_i , and $f'(x_i)$ is the function derivative at x_i .

The iteration formula for square root computation can be derived as follows:

1. Operation of sign: $S_q = 0$ (if $S_b = 1$ then square root is not possible).
2. Operation of exponent: $E_q = E_b + bias$.
3. Operation of Mantissa: square root of mantissa is calculated using NR method.

SquareRoot Algorithm

We perform NR square-root approximation using the function: $x_{i+1} = \frac{x_i}{2}(3 + dx_i^2)$. The steps are as follows:

1. NR approximation of x_n .
2. Partial Product
3. Addition
4. Normalization, to get IEEE754 floating format.

Newton Raphson Iteration is used to find the square root. It Uses 3 divide, 3 add and 2 Multiply Instances.

Square root Algorithm: $A^{0.5}$

A split into two parts $\Rightarrow M \times 2^E$

$$A^{0.5} = (M \times 2^E)^{0.5}$$

$$A^{0.5} = M^{0.5} \times 2^{(E/2)}$$

$$X = M^{0.5} \quad \text{and} \quad Z = 2^{(E/2)}$$

M adjusted to fit the range 0.5-1 by replacing exponent with 8'd126 (actual exponent = $126-127 = -1$).

$$X = (M \times 1)^{0.5}$$

$$X = (M \times 2^{(126-127)} / 2^{(126-127)})^{0.5}$$

$$X = (M \times 2^{(126-127)})^{0.5} / 2^{(-0.5)}$$

$$X = (M \times 2^{(126-127)})^{0.5} \times (1/2^{(-0.5)})$$

Let $C = 1/2^{(-0.5)}$ which is already known (constant) and multiplied at the end

Let $Y = (M \times 2^{(126-127)})^{0.5}$ which is computed using Newton Raphson Iterations and Inserted in the equation

thus X becomes: $X = Y \times C$

$2^{(E/2)}$ is basically exponent adjust and based on the value of E (Multiple of 2 or not). The resulting expression is multiplied by $2^{(0.5)}$ if the exponent is not a multiple of 2 and according to that values are re-adjusted at the end.

Initial Seed : $x_0 = 0.853553414345$

Newton Raphson Iterations :

$$x_1 = 0.5 \times (x_0 + X/x_0)$$

$$x_2 = 0.5 \times (x_1 + X/x_1)$$

$$x_3 = 0.5 \times (x_2 + X/x_2)$$

the exponent value of x_3 is adjusted and multiplied with square root of 2, if necessary to produce the final result.

Code Listing

Newton-Raphson (Full):

```
1  `timescale 1ns/1ps
2  module tst(
3      input [31:0]A,
4      input clk,
5      input reset,
6      output [31:0] f_sqrt
7  );
8      wire [7:0] A_exponent;
9      wire [22:0] A_mantissa;
10     wire A_sign;
11     wire [31:0] temp1,temp2,temp3,temp4,temp5,temp6,temp7,temp8,temp;
12     wire [31:0] x0,x1,x2,x3;
13     wire [31:0] sqrt_1by05; // 1/sqrt(0.5)
14     wire [31:0] sqrt_2; // sqrt(2)
15     wire [31:0] sqrt_1by2; // sqrt(0.5)
16     wire [7:0] Exp_2;
17     wire remainder;
18     wire pos;
19
20     // LookUp Values
21     assign x0 = 32'h3f5a827a;
22     assign sqrt_1by05 = 32'h3fb504f3;
23     assign sqrt_2 = 32'h3fb504f3;
24     assign sqrt_1by2 = 32'h3f3504f3;
25
26     // Number in IEEE754 Format
27     assign A_sign = A[31];
28     assign A_exponent = A[30:23];
29     assign A_mantissa = A[22:0];
30
31     // First Iteration
32     divide D1(.A({1'b0,8'd126,A_mantissa}),.B(x0),.C(temp1));
33     add A1(.A(temp1),.B(x0),.C(temp2));
34     assign x1 = {temp2[31],temp2[30:23]-1,temp2[22:0]};
35
36     // Second Iteration
37     divide D2(.A({1'b0,8'd126,A_mantissa}),.B(x1),.C(temp3));
38     add A2(.A(temp3),.B(x1),.C(temp4));
39     assign x2 = {temp4[31],temp4[30:23]-1,temp4[22:0]};
40
41     // Third Iteration
42     divide D3(.A({1'b0,8'd126,A_mantissa}),.B(x2),.C(temp5));
43     add A3(.A(temp5),.B(x2),.C(temp6));
44     assign x3 = {temp6[31],temp6[30:23]-1,temp6[22:0]};
45     multiply M1(.A(x3),.B(sqrt_1by05),.C(temp7));
46
47     assign pos = (A_exponent>=8'd127) ? 1'b1 : 1'b0;
48     assign Exp_2 = pos ? (A_exponent-8'd127)/2 : (A_exponent-8'd127-1)/2 ;
49     assign remainder = (A_exponent-8'd127)%2;
50     assign temp = {temp7[31],Exp_2 + temp7[30:23],temp7[22:0]};
51     //assign temp7[30:23] = Exp_2 + temp7[30:23];
52     multiply M2(.A(temp),.B(sqrt_2),.C(temp8));
53     assign f_sqrt = remainder ? temp8 : temp;
54
55 endmodule
56
57 module divide(
58     input [31:0]A,
59     input [31:0]B,
60     output [31:0]C
61 );
62     wire [7:0] exp_Brec;
63     wire [31:0] B_reciprocal;
```

```

64 wire [31:0] x0,x1,x2,x3;
65 wire [31:0] temp1,temp2,temp3,temp4,temp5,temp6,temp7,debug;
66
67 //Initial value
68 multiply M1(.A({1'b0,8'd126,B[22:0]}),.B(32'h3ff0f0f1),.C(temp1)); //verified
69 assign debug = {1'b1,temp1[30:0]};
70 add A1(.A(32'h4034b4b5),.B({1'b1,temp1[30:0]}),.C(x0));
71
72 //First Iteration
73 multiply M2(.A({1'b0,8'd126,B[22:0]}),.B(x0),.C(temp2));
74 add A2(.A(32'h40000000),.B({!temp2[31],temp2[30:0]}),.C(temp3));
75 multiply M3(.A(x0),.B(temp3),.C(x1));
76
77 //Second Iteration
78 multiply M4(.A({1'b0,8'd126,B[22:0]}),.B(x1),.C(temp4));
79 add A3(.A(32'h40000000),.B({!temp4[31],temp4[30:0]}),.C(temp5));
80 multiply M5(.A(x1),.B(temp5),.C(x2));
81
82 //Third Iteration
83 multiply M6(.A({1'b0,8'd126,B[22:0]}),.B(x2),.C(temp6));
84 add A4(.A(32'h40000000),.B({!temp6[31],temp6[30:0]}),.C(temp7));
85 multiply M7(.A(x2),.B(temp7),.C(x3));
86
87 //Reciprocal : 1/B
88 assign exp_Brec = x3[30:23]+8'd126-B[30:23];
89 assign B_reciprocal = {B[31],exp_Brec,x3[22:0]};
90
91 //Multiplication A*(1/B)
92 multiply M8(.A(A),.B(B_reciprocal),.C(C));
93
94 endmodule
95
96
97 module multiply (
98     input [31:0]A,
99     input [31:0]B,
100     output [31:0] C
101 );
102     reg [23:0] A_mantissa,B_mantissa;
103     reg [22:0] C_mantissa;
104     reg [47:0] Temp_mantissa;
105     reg [7:0] A_exponent,B_exponent,Temp_exponent,C_exponent;
106     reg A_sign,B_sign,C_sign;
107
108     always@(*) begin
109         A_mantissa = {1'b1,A[22:0]};
110         A_exponent = A[30:23];
111         A_sign = A[31];
112
113         B_mantissa = {1'b1,B[22:0]};
114         B_exponent = B[30:23];
115         B_sign = B[31];
116
117         Temp_exponent = A_exponent+B_exponent-127;
118         Temp_mantissa = A_mantissa*B_mantissa;
119         C_mantissa = Temp_mantissa[47] ? Temp_mantissa[46:24] : Temp_mantissa[45:23];
120         C_exponent = Temp_mantissa[47] ? Temp_exponent+1'b1 : Temp_exponent;
121         C_sign = A_sign^B_sign;
122     end
123     assign C = {C_sign,C_exponent,C_mantissa};
124
125 endmodule
126
127
128 module add(
129     input [31:0]A,
130     input [31:0]B,
131     output reg [31:0] C);

```

```

132
133 reg [23:0] a_mantissa,b_mantissa,temp_mantissa;
134 reg [22:0] C_mantissa;
135 reg [7:0] C_exponent,A_exponent,B_exponent,temp_exponent,diff_exponent;
136 reg C_sign,A_sign,B_sign,Temp_sign;
137 wire MSB;
138 reg [32:0] Temp;
139 reg carry,load;
140 reg comp;
141 reg [7:0] exp_adjust;
142 integer i;
143
144 always @(*) begin
145
146     comp = (A[30:23] >= B[30:23])? 1'b1 : 1'b0;
147
148     a_mantissa = comp ? {1'b1,A[22:0]} : {1'b1,B[22:0]};
149     A_exponent = comp ? A[30:23] : B[30:23];
150     A_sign = comp ? A[31] : B[31];
151
152     b_mantissa = comp ? {1'b1,B[22:0]} : {1'b1,A[22:0]};
153     B_exponent = comp ? B[30:23] : A[30:23];
154     B_sign = comp ? B[31] : A[31];
155
156     diff_exponent = A_exponent-B_exponent;
157     b_mantissa = (b_mantissa >> diff_exponent);
158     {carry,temp_mantissa} = (A_sign ^ B_sign)? a_mantissa + b_mantissa : a_mantissa-
b_mantissa ;
159     exp_adjust = A_exponent;
160     if(carry) begin
161         temp_mantissa = temp_mantissa>>1;
162         exp_adjust = exp_adjust+1'b1;
163     end else begin
164         load =0;
165         for (i=0;i<24;i=i+1) begin
166             if (temp_mantissa[23] ==0 && load ==0) begin
167                 temp_mantissa = temp_mantissa<<1;
168                 exp_adjust = exp_adjust-1'b1;
169             end else load =1;
170         end
171     end
172
173     C_sign = A_sign;
174     C_mantissa = temp_mantissa[22:0];
175     C_exponent = exp_adjust;
176     C= {C_sign,C_exponent,C_mantissa};
177
178 end
179
180 endmodule

```

TestBench:

```
1  'timescale 1ns/1ps
2  module tsttb();
3      reg [31:0] A;
4      wire [31:0] f_sqrt;
5      tst s1 (.A(A),.f_sqrt(f_sqrt)
6  );
7      initial begin
8          $monitor("A=%h, f_sqrt=%h", A, f_sqrt);
9          A = 32'h42040000; // 33
10         #20 A = 32'h42aa0000; // 85
11         #20 A = 32'h42b80000; // 92
12         #20 A = 32'h44208000; // 642
13         #20 A = 32'h4517f000; // 2431
14         #20 $finish;
15     end
16
17 endmodule
```


Results

| | | 100.000 ns | | | | |
|----------------|----------|------------|-----------|-----------|-----------|-----------|
| Name | Value | 0.000 ns | 20.000 ns | 40.000 ns | 60.000 ns | 80.000 ns |
| > A[31:0] | 4517f000 | 42040000 | 42aa0000 | 42b80000 | 44208000 | 4517f000 |
| > f_sqrt[31:0] | 4245387d | 40b7d375 | 41138340 | 41197774 | 41cab3a6 | 4245387d |

Here values are in hexadecimal format of IEEE754 standard, but when we convert it to decimal number system it results in:

| | | 100.000 ns | | | | |
|----------------|----------------|------------|-----------|-----------|-----------|-----------|
| Name | Value | 0.000 ns | 20.000 ns | 40.000 ns | 60.000 ns | 80.000 ns |
| > A[31:0] | 2431.0 | 33.0 | 85.0 | 92.0 | 642.0 | 2431.0 |
| > f_sqrt[31:0] | 49.30619812011 | 5.7448997 | 9.2196187 | 9.5916986 | 25.338054 | 49.306198 |

| Input (IEEE754) (Hexadecimal representation) | Input (Decimal Number System) | Output (IEEE754) (Hexadecimal representation) | Output (Decimal Number System) | Actual square root |
|---|--|--|--------------------------------------|--------------------|
| 42040000 | 33 | 40b7d375 | 5.74456262589 | 5.74456264653 |
| 42aa0000 | 85 | 41138340 | 9.21954345703 | 9.21954445729 |
| 42b80000 | 92 | 41197774 | 9.5916633606 | 9.59166304662 |
| 44208000 | 642 | 41cab3a6 | 25.3377189636 | 25.33771891863 |
| 4517f000 | 2431 | 4245387d | 49.3051643372 | 49.30517214248 |

Simulation Results

| Tcl Console | | Messages | | Log | | Reports | | Design Runs | | Utilization | | x | |
|-----------------------|--|----------|--|---------------|--|---------|--|-------------|--|-------------|--|----|--|
| Q | | ≡ | | ⌵ | | ⏏ | | Q | | ≡ | | ⌵ | |
| | | | | | | | | % | | Hierarchy | | | |
| Hierarchy | | | | | | | | | | | | | |
| Summary | | | | | | | | | | | | | |
| ▼ Slice Logic | | ▼ N | | tst | | | | 6811 | | 1 | | 52 | |
| ▼ Slice LUTs (5%) | | | | | | | | | | | | | |
| LUT as Logic (5%) | | | | | | | | | | | | | |
| F7 Muxes (<1%) | | | | | | | | | | | | | |
| Memory | | > | | D1 (divide) | | | | 1741 | | 0 | | 16 | |
| ▼ DSP | | > | | D2 (divide_2) | | | | 1857 | | 1 | | 16 | |
| ▼ DSPs (7%) | | > | | D3 (divide_3) | | | | 1815 | | 0 | | 16 | |
| DSP48E1 only | | | | | | | | | | | | | |
| M1 (multiply) | | | | | | | | 154 | | 0 | | 2 | |
| ▼ IO and GT Specific | | | | | | | | | | | | | |
| Bonded IOB (22%) | | | | | | | | | | | | | |
| Clocking | | | | | | | | | | | | | |
| Specific Feature | | | | | | | | | | | | | |
| Primitives | | | | | | | | | | | | | |
| Black Boxes | | | | | | | | | | | | | |
| Instantiated Netlists | | | | | | | | | | | | | |

Utilization Results

Conclusions

- The implemented design is fully compatible with IEEE754 standard. Using the Newton/Raphson method, the design is fast, and has high precision.
- The implemented block can be used for implementing multiplier/divider blocks for and single-precision floating pt as well.

Appendix

Work done prior to mid-semester.

Floating Pt Square-Root Implementation

Midsem Report

Abhishek Verma (214102401)
Chaitanya Tejaswi (214102408)
Pawan Kumar (214102412)

20-04-2022

Aim

Perform post-synthesis simulations & functionality testing for Floating Pt Square-Root Implementation.

Implementation#1: 32bit Integer Square-Root

Approach

We perform restoring Restoring approach. Examples for decimal/binary division this way are given below:

Decimal

| | q_1 | q_2 | q_3 | q_4 | q_5 | q_6 | |
|-------------|--------|-------|-------|-------|-------|-------|---|
| | 1 | 7 | 3 | 2 | 0 | 5 | ; Square root |
| 1 | √ | 03 | 00 | 00 | 00 | 00 | ; Radicand |
| $(q_1 = 1)$ | 1 | 1 | | | | | ; $1 \times 1 = 1$ (guess 1) |
| | 27 | 2 | 00 | | | | ; $1 + 1 = 2$ |
| $(q_2 = 7)$ | 7 | 1 | 89 | | | | ; $27 \times 7 = 189$ (guess 7) |
| | 343 | 11 | 00 | | | | ; $27 + 7 = 34$ |
| $(q_3 = 3)$ | 3 | 10 | 29 | | | | ; $343 \times 3 = 1029$ (guess 3) |
| | 3462 | 71 | 00 | | | | ; $343 + 3 = 346$ |
| $(q_4 = 2)$ | 2 | 69 | 24 | | | | ; $3462 \times 2 = 6924$ (guess 2) |
| | 34640 | 1 | 76 | 00 | | | ; $3462 + 2 = 3464$ |
| $(q_5 = 0)$ | 0 | | | 0 | | | ; $34640 \times 0 = 0$ (guess 0) |
| | 346405 | 1 | 76 | 00 | 00 | | ; $34640 + 0 = 34640$ |
| $(q_6 = 5)$ | 5 | 1 | 73 | 20 | 25 | | ; $346405 \times 5 = 1732025$ (guess 5) |
| | | 2 | 79 | 75 | | | ; Remainder |

Binary

| | q_1 | q_2 | q_3 | q_4 | |
|-------------|-------|-------|-------|-------|--------------------------------------|
| | 1 | 0 | 1 | 1 | ; Square root (11_{10}) |
| 1 | √ | 01 | 11 | 11 | ; Radicand (127_{10}) |
| $(q_1 = 1)$ | 1 | 1 | | | ; $1 \times 1 = 1$ (guess 1) |
| | 100 | 0 | 11 | | ; $1 + 1 = 10$ |
| $(q_2 = 0)$ | 0 | 0 | | | ; $100 \times 0 = 0$ (guess 0) |
| | 1001 | 11 | 11 | | ; $100 + 0 = 100$ |
| $(q_3 = 1)$ | 1 | 10 | 01 | | ; $1001 \times 1 = 1001$ (guess 1) |
| | 10101 | 1 | 10 | 11 | ; $1001 + 1 = 1010$ |
| $(q_4 = 1)$ | 1 | 1 | 01 | 01 | ; $10101 \times 1 = 10101$ (guess 1) |
| | | 01 | 10 | | ; Remainder (6_{10}) |

Implementation#2: 32bit Floating Pt Square-Root

Approach

We perform NR square-root approximation using the function: $x_{i+1} = \frac{x_i}{2}(3 + dx_i^2)$. The steps are as follows:

1. NR approximation of x_n .
2. Partial Product
3. Addition
4. Normalization, to get IEEE754 floating format.

Newton/Raphson SquareRoot Approximation

$$1.) \sqrt{0.25} = 0.5$$

$$\sqrt{0x0.40000000} = \sqrt{\frac{4}{16}} = 0x0.80000000 = \frac{8}{16} = 0.5$$

$$2.) \sqrt{0.75} = 0.8660254038$$

$$\sqrt{0x0.c0000000} = \sqrt{\frac{12}{16}} = 0x0.ddb3d743 = \frac{13}{16^1} + \frac{13}{16^2} + \frac{11}{16^3} + \frac{3}{16^4} + \frac{13}{16^5} + \frac{7}{16^7} + \frac{4}{16^4} + \frac{4}{16^3} = 0.8660254038$$

(correct upto 12 decimal places).

Let d such that $\frac{1}{4} \leq d < 1$, ie: $(= 0.1xxx..., 0.01xxx..., 0.001xxx..., \dots)$

Calculate $q = \sqrt{d}$


We calculate $x_n = \frac{1}{\sqrt{d}}$, then obtain $q = dx_n = d \frac{1}{\sqrt{d}} = \sqrt{d}$

Let $f(x) = \frac{1}{x^2} - d$, then $f(x) = 0$ for $x = \frac{1}{\sqrt{d}}$.


$$\begin{aligned} x_{i+1} &= x_i - \frac{f(x_i)}{f'(x_i)} \\ &= x_i - \frac{\frac{1}{x_i^2} - d}{-\frac{2}{x_i^3}} \\ &= \frac{x_i}{2} (3 + dx_i^2) \end{aligned}$$

Procedure


1. Create new project in Vivado, and source appropriate files for each (one for logic, one for testbench).
2. Run behavioral simulation, and RTL analysis/synthesis/implementation.
3. Using the Tcl commands **report_timing_summary** & **report_power**, note down the delay (slack), power consumption, and no. of LUTs/FFs used for each implementation.



Implementation 1: Restoring SquareRoot



Implementation 2: Newton-Raphson (Basic) SquareRoot



Implementation 2: Newton-Raphson (Full) SquareRoot

Results

| . | LUTs | FFs | Delay | Power |
|------------------------|------|-----|----------|---------|
| 32-Bit Integer Sqrt | 111 | 12 | 13.187ns | 3.326W |
| 32-Bit FloatingPt Sqrt | 1326 | 54 | 22.47ns | 16.475W |

Table 1: Timing/Power Values for Implementations

Code Listing

Restoring:

```
1  `timescale 1ns/1ns
2  module root_restoring (d,load,clk,reset,q,r,busy,ready,count);
3      input  [31:0] d;                // Input
4      input          load;
5      output reg     busy, ready;
6      input          clk, reset;
7      output [15:0] q;                // Root
8      output [16:0] r;                // Remainder
9      output [3:0] count;
10     reg  [31:0] reg_d;
11     reg  [15:0] reg_q;
12     reg  [16:0] reg_r;
13     reg  [3:0] count;
14     wire [17:0] sub_out = {reg_r[15:0],reg_d[31:30]} - {reg_q,2'b1}; // -
15     wire [16:0] rem_out = sub_out[17]?                // restoring
16                     {reg_r[14:0],reg_d[31:30]} : sub_out[16:0]; // or not
17     assign q = reg_q;
18     assign r = reg_r;
19     always @(posedge clk or negedge reset) begin
20         if (!reset) begin
21             busy  <= 0; ready <= 0;
22         end else begin
23             if (load) begin
24                 reg_d <= d; reg_q <= 0; reg_r <= 0;
25                 busy  <= 1; ready <= 0; count <= 0;
26             end else if (busy) begin
27                 // Next-State Logic
28                 reg_d <= {reg_d[29:0],2'b0};
29                 reg_q <= {reg_q[14:0],~sub_out[17]};
30                 reg_r <= rem_out;
31                 count <= count + 4'b1;
32                 if (count == 4'hf) begin                // max 16 iterations
33                     busy  <= 0; ready <= 1;            // done! q,r ready
34                 end
35             end
36         end
37     end
38 endmodule
```

TestBench:

```
1  `timescale 1ns/1ns
2  module root_restoring_tb;
3      reg [31:0] d;
4      reg      load;
5      reg      clk,reset;
6      wire [15:0] q;
7      wire [16:0] r;
8      wire      busy;
9      wire      ready;
10     wire [3:0] count;
11     root_restoring rt (d,load,clk,reset,q,r,busy,ready,count);
12     initial begin
13         $dumpfile ("root_restoring.vcd");
14         $dumpvars (0, root_restoring);
15         $monitor ("At ", $time, ": D=%d, Root=%d, Remainder=%d",d,q,r);
16         reset = 0; load = 0; clk = 1; d = 32'h0000007f;
17         #5 reset = 1; load = 1;
18         #10 load = 0;
19         #250 reset = 0; load = 0; clk = 1; d = 32'h000000c4;
20         #5 reset = 1; load = 1;
21         #10 load = 0;
22         #250 reset = 0; load = 0; clk = 1; d = 32'h000000e3;
23         #5 reset = 1; load = 1;
24         #10 load = 0;
25         #250 reset = 0; load = 0; clk = 1; d = 32'h000000e3;
26         #5 reset = 1; load = 1;
27         #10 load = 0;
28         #250 \ $finish;
29     end
30     always #5 clk = !clk;
31 endmodule
```

Newton-Raphson (Basic):

```

1 module root_newton (d,clk,reset,start,q,busy,ready,count);
2     input  [31:0] d;           // Input
3     input    clk, reset;
4     input    start;
5     output [31:0] q;           // Root
6     output reg  busy, ready;
7     output [2:0] count;
8     reg  [31:0] reg_d;
9     reg  [33:0] reg_x;
10    reg  [1:0] count;
11    // Partial Sums for:  $X[i+1] = X_i * (3 - X_i * X_i * d) / 2$ 
12    // X2, X2d, (3-X2d), X(3-X2d), Xd, X0
13    wire [67:0] X2      = reg_x * reg_x;
14    wire [67:0] X2d     = reg_d * X2[67:32];
15    wire [33:0] k_X2d   = 34'h300000000 - X2d[65:32];
16    wire [67:0] Xk_X2d  = reg_x * k_X2d;
17    wire [65:0] Xd      = reg_d * reg_x;
18    wire  [7:0] X0      = lut(d[31:27]);
19    assign      q       = Xd[63:32] + |Xd[31:0]; // RoundOff
20    always @(posedge clk or negedge reset) begin
21        if (!reset) begin
22            busy <= 0; ready <= 0;
23        end else begin
24            if (start) begin
25                reg_d <= d; reg_x <= {2'b1,X0,24'b0};
26                busy <= 1; ready <= 0; count <= 0;
27            end else begin
28                reg_x <= Xk_X2d[66:33]; count <= count + 2'b1;
29                if (count == 2'h2) begin // max 3 iterations
30                    busy <= 0; ready <= 1; // done! q,r ready
31                end
32            end
33        end
34    end
35    function [7:0] lut; // 1/sqrt(d)
36        input [4:0] d;
37        case (d)
38            5'h08: lut = 8'hff;
39            5'h09: lut = 8'he1;
40            5'h0a: lut = 8'hc7;
41            5'h0b: lut = 8'hb1;
42            5'h0c: lut = 8'h9e;
43            5'h0d: lut = 8'h9e;
44            5'h0e: lut = 8'h7f;
45            5'h0f: lut = 8'h72;
46            5'h10: lut = 8'h66;
47            5'h11: lut = 8'h5b;
48            5'h12: lut = 8'h51;
49            5'h13: lut = 8'h48;
50            5'h14: lut = 8'h3f;
51            5'h15: lut = 8'h37;
52            5'h16: lut = 8'h30;
53            5'h17: lut = 8'h29;
54            5'h18: lut = 8'h23;
55            5'h19: lut = 8'h1d;
56            5'h1a: lut = 8'h17;
57            5'h1b: lut = 8'h12;
58            5'h1c: lut = 8'h0d;
59            5'h1d: lut = 8'h08;
60            5'h1e: lut = 8'h04;
61            5'h1f: lut = 8'h00;
62            default: lut = 8'hff;
63        endcase
64    endfunction
65 endmodule

```

TestBench:

```
1  'timescale 1ns/1ns
2  module root_newton_tb;
3      reg    [31:0] d;
4      reg          start;
5      reg          clk,reset;
6      wire [31:0] q;
7      wire          busy;
8      wire          ready;
9      wire  [1:0] count;
10     root_newton root (d,clk,reset,start,q,busy,ready,count);
11     initial begin
12         $dumpfile ("root_newton.vcd");
13         $dumpvars (0, root_newton);
14         $monitor ("At ", $time, ": D=0.%h, Root=0.%h",d,q);
15
16         // d = 0x0.40000000 = 0.25
17         // sqrt(d) = 0x0.80000000 = 0.5
18         reset=0; start=0; clk=1; d=32'h40000000;
19         #35 reset=1; start=1;
20         #70 start= 0;
21
22         // d = 0x0.c0000000 = 0.75
23         // sqrt(d) = 0x0.ddb3d743 = 0.8660254038
24         #455 d      = 32'hc0000000;
25         #35 start= 1;
26         #70 start= 0;
27
28         // d = 0x0.f8000000 = 0.96875
29         // sqrt(d) = 0.9842509843 = 0x0.fbe00000
30         #455 d      = 32'hf8000000;
31         #35 start= 1;
32         #70 start= 0;
33
34         #455 \ $finish;
35     end
36     always #35 clk = !clk;
37 endmodule
```